

Specification of the Amy language

Mutable Variables & Constant Propagation

Computer Language Processing

LARA

Autumn 2019

1 Syntax

The syntax of Amy is given formally by the context-free grammar of Figure 1. Everything spelled in *italic* is a nonterminal symbol of the grammar, whereas the terminal symbols are spelled in **monospace** font. `*` is the Kleene star, s^+ stands for one or more repetitions of s , and $?$ stands for optional presence of a symbol (zero or one repetitions). The square brackets `[]` are not symbols of the grammar, they merely group symbols together.

Before parsing an Amy program, the Amy *lexer* generates a sequence of terminal symbols (*tokens*) from the source files. Some nonterminal symbols mentioned, but not specified, in Figure 1 are also represented as a single token by the lexer. They are lexed according to the rules in Figure 2. In Figure 2, we denote the range between characters α and β (included) with $[\alpha - \beta]$.

The syntax in Figure 1 is an *overapproximation* of the real syntax of Amy. This means that it allows some programs that should not be allowed in Amy. To get the real syntax of Amy, there are some additional restrictions presented (among other things) in the following notes:

- The reserved words of Amy are the following: **abstract**, **Boolean**, **case**, **class**, **def**, **else**, **error**, **extends**, **false**, **if**, **Int**, **match**, **object**, **String**, **true**, **Unit**, **val**, **var**, `_` (the wildcard pattern).

Identifiers are not allowed to coincide with a reserved word.

- The operators and language constructs of Amy have the following precedence, starting from the *lowest*:

(1) **val**, **var**; (2) **if**, **match** (3) `||` (4) **&&** (5) **==** (6) **<**, **<=** (7) **+**, **-**, **++** (8) *****, **/**, **%** (9) Unary **-**, **!** (10) **error**, calls, variables, literals, parenthesized expressions.

```

Program ::= Module*
Module ::= object Id { Definition* Expr? }
Definition ::= AbstractClassDef | CaseClassDef | FunDef
AbstractClassDef ::= abstract class Id
CaseClassDef ::= case class Id ( Params ) extends Id
FunDef ::= def Id ( Params ) : Type = { Expr }
Params ::=  $\epsilon$  | ParamDef [ , ParamDef ]*
ParamDef ::= Id : Type
Type ::= Int | String | Boolean | Unit | [ Id . ]? Id
Expr ::= Id
      | Id = Expr
      | Literal
      | Expr BinOp Expr
      | UnaryOp Expr
      | [ Id . ]? Id ( Args )
      | Expr ; Expr
      | ( val | var ) ParamDef = Expr ; Expr
      | if ( Expr ) { Expr } else { Expr }
      | Expr match { MatchCase+ }
      | error ( Expr )
      | ( Expr )
Literal ::= true | false | ( )
BinOp ::= + | - | * | / | % | < | <=
        | && | || | == | ++
UnaryOp ::= - | !
MatchCase ::= case Pattern => Expr
Pattern ::= [ Id . ]? Id ( Patterns ) | Id | Literal | _
Patterns ::=  $\epsilon$  | Pattern [ , Pattern ]*
Args ::=  $\epsilon$  | Expr [ , Expr ]*

```

Figure 1: Syntax of Amy

```

IntLiteral ::= Digit+
Id ::= Alpha AlphaNum* (and not a reserved word)
AlphaNum ::= Alpha | Digit | _
Alpha ::= [a - z] | [A - Z]
Digit ::= [0 - 9]
StringLiteral ::= " StringChar* "
StringChar ::= Any character except newline and "

```

Figure 2: Lexical rules for Amy

For example,

`1 + 2 * 3` means `1 + (2 * 3)` and

`1 + 2 match {...}` means `(1 + 2) match {...}`.

A little more complicated is the interaction between `;` and `val`: the definition part of the `val` extends only as little as the first semicolon, but then the variable defined is visible through any number of semicolons. Thus `(val x: Int = y; z; x)` means `(val x: Int = y; (z; x))` and not `(val x: Int = (y; z); x)` or `((val x: Int = y; z); x)` (i.e. `x` takes the value of `y` and is visible until the end of the expression).

All operators are left-associative. That means that within the same precedence category, the leftmost application of an operator takes precedence. An exception is the sequence operator, which for ease of the implementation (you will understand during parsing) can be considered right-associative (it is an associative operator so it does not really matter).

- A `val` (or `var`) definition is not allowed directly in the value assigned by an enclosing `val` definition. E.g. `(val x: Int = val y: Int = 0; 1; 2)` is not allowed. On the other hand, `(val x: Int = 0; val y: Int = 1; 2)` is allowed.
- It is not allowed to use a `val` (or `var`) as a (second) operand to an operator. E.g. `(1 + val x: Int = 2; x)` is not allowed.
- A unary operator is not allowed as a direct argument of another unary operator. E.g. `--x` is not allowed.
- It is not allowed to use `match` as a first operand of any binary operator, except `;`. E.g. `(x match { ... } + 1)` is not allowed. On the other hand `(x match { ... }; x)` is allowed.
- The syntax `[Id .]? Id` refers to an optionally qualified name, for example either `MyModule.foo` or `foo`. If the qualifier is included, the qualified name refers to a definition `foo` in another module `MyModule`; otherwise, `foo` should be defined in the current module. Since Amy does not have the import statement of Scala or Java, this is the only way to refer to definitions in other modules.
- One line comments are introduced with `//`: *//This is a comment.* Everything until the end of the line is a comment and should be ignored by the lexer.
- Multiline comments can be used as follows: */*This is a comment */.* Everything between the delimiters is a comment, notably including new-line characters and `/*`. Nested comments are not allowed.
- Escaped characters are not recognised inside string literals. I.e. `"\n"` stands for a string literal which contains a backspace and an `"n"`.
- Variable assignments are to return the value of the variable after the assignment.

1.1 Implementation of changes

I added the keyword `var` and added to the grammar the modifications that appear `red` in the lexical rules. For that I had to create an auxiliary class to obey the Scalion rules that prevent logical Or's from returning non-matching types.

I added the new class `Assignment` to every file of the pipeline, as defined in the specification in `red`.

2 Type checking

2.1 Implementation of changes

In the AST, local definitions contain whether the variables are mutable or immutable. During type checking we must also store to the environment the mutability of the local variables. Down below you can find the set of rules followed, showing how types are considered as Cartesian products (Type, Mutability).

VARIABLE $\frac{v : (T_1, T_2) \in \Gamma}{\Gamma \vdash v : (T_1, T_2)}$	INT LITERAL $\frac{i \text{ is an integer literal}}{\Gamma \vdash i : (\text{Int}, \text{I})}$	STRING LITERAL $\frac{s \text{ is a string literal}}{\Gamma \vdash s : (\text{String}, \text{I})}$
UNIT $\frac{}{\Gamma \vdash () : (\text{Unit}, \text{I})}$	BOOLEAN LITERAL $\frac{b \in \{\text{true}, \text{false}\}}{(\Gamma \vdash b : \text{Boolean}, \text{I})}$	
ARITH. BIN. OPERATORS $\frac{\Gamma \vdash e_1 : (\text{Int}, \text{Any}) \quad \Gamma \vdash e_2 : (\text{Int}, \text{Any}) \quad op \in \{+, -, *, /, \%\}}{\Gamma \vdash e_1 \text{ op } e_2 : (\text{Int}, \text{I})}$		
ARITH. COMP. OPERATORS $\frac{\Gamma \vdash e_1 : (\text{Int}, \text{Any}) \quad \Gamma \vdash e_2 : (\text{Int}, \text{Any}) \quad op \in \{<, <=\}}{\Gamma \vdash e_1 \text{ op } e_2 : (\text{Boolean}, \text{I})}$		
ARITH. NEGATION $\frac{\Gamma \vdash e : (\text{Int}, \text{Any})}{\Gamma \vdash -e : (\text{Int}, \text{I})}$		
BOOLEAN BIN. OPERATORS $\frac{\Gamma \vdash e_1 : (\text{Boolean}, \text{Any}) \quad \Gamma \vdash e_2 : (\text{Boolean}, \text{Any}) \quad op \in \{\&\&, \}}{\Gamma \vdash e_1 \text{ op } e_2 : (\text{Boolean}, \text{I})}$		
BOOLEAN NEGATION $\frac{\Gamma \vdash e : (\text{Boolean}, \text{Any})}{\Gamma \vdash !e : (\text{Boolean}, \text{I})}$	STRING CONCATENATION $\frac{\Gamma \vdash e_1 : (\text{String}, \text{I}) \quad \Gamma \vdash e_2 : (\text{String}, \text{I})}{\Gamma \vdash e_1 ++ e_2 : (\text{String}, \text{I})}$	
EQUALITY $\frac{\Gamma \vdash e_1 : (T, \text{Any}) \quad \Gamma \vdash e_2 : (T, \text{Any})}{\Gamma \vdash e_1 == e_2 : (\text{Boolean}, \text{I})}$		
SEQUENCE $\frac{\Gamma \vdash e_1 : (T_1, \text{Any}) \quad \Gamma \vdash e_2 : (T_2, T)}{\Gamma \vdash e_1 ; e_2 : (T_2, T)}$	ASIGNATION $\frac{\Gamma \vdash e_1 : (T_1, \text{M}) \quad \Gamma \vdash e_2 : (T_1, \text{Any})}{\Gamma \vdash e_1 = e_2 : (T_1, \text{M})}$	
LOCAL IMMUTABLE VARIABLE DEFINITION $\frac{\Gamma \vdash e_1 : T_1 \quad \Gamma, n : (T_1, \text{I}) \vdash e_2 : T_2}{\Gamma \vdash \text{val } n : T_1 = e_1 ; e_2 : T_2}$		
LOCAL MUTABLE VARIABLE DEFINITION $\frac{\Gamma \vdash e_1 : (T_1, \text{Any}) \quad \Gamma, n : (T_1, \text{M}) \vdash e_2 : T_2}{\Gamma \vdash \text{var } n : T_1 = e_1 ; e_2 : T_2}$		

Figure 3: Typing rules for expressions

$$\begin{array}{c}
\text{FUNCTION/CLASS CONSTRUCTOR INVOCATION} \\
\frac{\Gamma \vdash e_1 : (T_1, \mathbf{Any}) \quad \dots \quad \Gamma \vdash e_n : (T_n, \mathbf{Any}) \quad \Gamma \vdash f : (T_1, \dots, T_n) \Rightarrow T}{\Gamma \vdash f(e_1, \dots, e_n) : (T, \mathbf{I})} \\
\\
\begin{array}{cc}
\text{IF-THEN-ELSE} & \text{ERROR} \\
\frac{\Gamma \vdash e_1 : (\mathbf{Boolean}, \mathbf{Any}) \quad \Gamma \vdash e_2 : T \quad \Gamma \vdash e_3 : T}{\Gamma \vdash \mathbf{if} (e_1) \{e_2\} \mathbf{else} \{e_3\} : T} & \frac{\Gamma \vdash e : (\mathbf{String}, \mathbf{Any})}{\Gamma \vdash \mathbf{error}(e) : T}
\end{array} \\
\\
\text{PATTERN MATCHING} \\
\frac{\Gamma \vdash e : (T_s, \mathbf{Any}) \quad \forall i \in [1, n]. \Gamma \vdash p_i : (T_s, \mathbf{I}) \quad \forall i \in [1, n]. \Gamma, \mathit{bindings}(p_i) \vdash e_i : T_c}{\Gamma \vdash e \mathbf{match} \{ \mathbf{case} p_1 \Rightarrow e_1 \dots \mathbf{case} p_n \Rightarrow e_n \} : T_c}
\end{array}$$

Figure 4: Typing rules for expressions

2.2 Constant propagation

I implemented the following simplifications.

IF-THEN-ELSE TRUE $\frac{e_1 \rightsquigarrow \text{True}}{\text{if } (e_1) \{e_2\} \text{ else } \{e_3\} \rightsquigarrow e_2}$		IF-THEN-ELSE FALSE $\frac{e_1 \rightsquigarrow \text{False}}{\text{if } (e_1) \{e_2\} \text{ else } \{e_3\} \rightsquigarrow e_3}$	
BOOLEAN SYMMETRY $\frac{\text{op} \in \{ , \&\&\}}{e_1 \text{ op } e_2 \rightsquigarrow e_2 \text{ op } e_1}$	AND 1 $\frac{e_1 \rightsquigarrow \text{False}}{e_1 \&\& e_2 \rightsquigarrow \text{False}}$	AND 2 $\frac{e_1 \rightsquigarrow \text{True}}{e_1 \&\& e_2 \rightsquigarrow e_2}$	
OR 1 $\frac{e_1 \rightsquigarrow \text{False}}{e_1 e_2 \rightsquigarrow e_2}$	OR 2 $\frac{e_1 \rightsquigarrow \text{True}}{e_1 \&\& e_2 \rightsquigarrow \text{True}}$	IF-THEN-ELSE TRUE $\frac{e_1 \rightsquigarrow \text{False}}{\text{if } (e_1) \{e_2\} \text{ else } \{e_3\} \rightsquigarrow e_3}$	
INT BINARY OP $\frac{e_1 \rightsquigarrow \mathbf{n} \quad e_2 \rightsquigarrow \mathbf{m}}{e_1 \text{ op } e_2 \rightsquigarrow \mathbf{n} \text{ op } \mathbf{m}}$		DEFINITIONS $\frac{\text{va* } x = e_1; e_2 \quad e_1 \rightsquigarrow e_1^*}{e_2 \rightsquigarrow \text{subs}(e_2, x, e_1^*)}$	
FUNCTION $\frac{e_i \rightsquigarrow e_i^*}{f(e_1, \dots, e_i, \dots, e_n) \rightsquigarrow f(e_1, \dots, e_i^*, \dots, e_n)}$			

Figure 5: Constant propagation

2.3 Implementation of changes

We move through the code remembering the values of variables and in-lining them when possible. All is performed in the file `ConstantPropagation.scala`. There you can find how it is done following a scheme that closely resembles the other steps of the pipeline.

3 Further information

All further information is commented on each file. On the respective position of the change.